A noncommutatively rigid Fano fivefold

Pieter Belmans

September 9, 2026

Abstract

We construct a Fano fivefold that has vanishing second (and first) Hochschild cohomology. Its category of coherent sheaves is therefore rigid: it has no infinitesimal deformations whatsoever.

We work over the field $\mathbb{C}$. Consider the multiprojective space $X = \mathbb{P}^2_{\mathbf{a}} \times \mathbb{P}^2_{\mathbf{b}} \times \mathbb{P}^2_{\mathbf{c}} \times \mathbb{P}^3_{\mathbf{d}}$, and the complete intersection Z given by

$$
\begin{aligned}
a_0b_0 + a_1b_1 + a_2b_2 &= 0 \\
b_0d_0 + b_1d_3 + b_2(d_1 + d_2) &= 0 \\
c_0d_0 + c_1d_1 + c_2d_2 &= 0 \\
c_0d_1 + c_1d_2 + c_2d_3 &= 0.
\end{aligned}
\tag{1}
$$

These equations define a section s of the vector bundle $\mathcal{E} := \mathcal{O}_X(1,1,0,0) \oplus \mathcal{O}_X(0,1,0,1) \oplus \mathcal{O}_X(0,0,1,1)^{\oplus 2}$. The purpose of this note is to collect the following properties of Z.

Theorem A. *The complete intersection Z is a smooth projective Fano fivefold, with only diagonal non-zero Hodge numbers given by $(1,4,8,8,4,1)$, for which*

$$
\mathrm{H}^0(Z, \mathrm{T}_Z) = \mathrm{H}^1(Z, \mathrm{T}_Z) = \mathrm{H}^0(Z, \bigwedge^2 \mathrm{T}_Z) = 0. \tag{2}
$$

Using the Hochschild–Kostant–Rosenberg decomposition for Hochschild cohomology and Kodaira vanishing, we see that (2) is equivalent to $\mathrm{HH}^1(Z) = \mathrm{HH}^2(Z) = 0$.

We will prove this using computer algebra, in particular we will use Macaulay2 [5] for computations with explicit coordinates, and PartialFlagVarieties.jl [3] for computations with zero loci of vector bundles. The code is available in [1]. It is classical that there are no examples of del Pezzo surfaces with vanishing second Hochschild cohomology, and for Fano 3-folds this follows from [2]. We have not found an example for Fano 4-folds.

Smooth Fano fivefold We first verify that the equations in (1) define a nonempty smooth complete intersection of codimension 4 in X. Let $S = \mathbb{Q}[\mathbf{a},\mathbf{b},\mathbf{c},\mathbf{d}]$ be the polynomial ring with the $\mathbb{Z}^4$-grading determined by the four factors of X, and let $I \subset S$ be the ideal generated by these equations. The following Macaulay2 code checks that Z has dimension 5 and is smooth.

```
needsPackage "MultiprojectiveVarieties";

S = QQ[a0,a1,a2, b0,b1,b2, c0,c1,c2, d0,d1,d2,d3,
    Degrees => {3:{1,0,0,0}, 3:{0,1,0,0},
                3:{0,0,1,0}, 4:{0,0,0,1}}];
I = ideal {
    a0*b0 + a1*b1 + a2*b2,
    b0*d0 + b1*d3 + b2*(d1+d2),
    c0*d0 + c1*d1 + c2*d2,
    c0*d1 + c1*d2 + c2*d3
};

Z = projectiveVariety I;
assert(dim Z == 5);
assert(dim singularLocus Z == -1);
```

By adjunction, the anticanonical bundle is

$$(3)\qquad \omega_Z^\vee \cong (\omega_X^\vee \otimes (\det \mathcal{E})^\vee)|_Z \cong \mathcal{O}_Z(3-1, 3-2, 3-2, 4-3) = \mathcal{O}_Z(2,1,1,1).$$

Since $\mathcal{O}_X(2,1,1,1)$ is ample, its restriction to Z is ample, so Z is a Fano fivefold.

Cohomology of bundles The following Julia code computes the Hodge numbers in Theorem A, and computes the cohomologies of bundles we will use in what comes next.

```
using PartialFlagVarieties

X = product(projective_space(2), projective_space(2),
            projective_space(2), projective_space(3))
E = O(X, [1,1,0,0]) + O(X, [0,1,0,1]) + 2 * O(X, [0,0,1,1])
Z = zero_locus(E)

diagonal = [1,4,8,8,4,1]
@assert hodge_numbers(Z) == [p == q ? diagonal[p+1] : 0
                              for p in 0:5, q in 0:5]

T_XZ = restrict(Z, tangent_bundle(X))
N_ZX = normal_bundle(Z)
for (bundle, h0) in [(T_XZ, 39), (N_ZX, 39),
                     (exterior_power(T_XZ, 2), 582), (T_XZ * N_ZX, 1083)]
    H = cohomology(bundle)
    @assert H[0] == h0
    @assert all(H[q] == 0 for q in 1:5)
end

H = cohomology(tangent_bundle(Z))
@assert H[0] == H[1]
@assert all(H[q] == 0 for q in 2:5)
```

Setting up vanishing We now describe these bundles and the sequences relating them, so that we can verify (2). Let e_i denote the ith basis vector and write $(n_1, n_2, n_3, n_4) = (2,2,2,3)$. The restriction of the Euler sequence on X to the zero locus Z is

$$(4)\qquad 0 \to \mathcal{O}_Z^{\oplus 4} \xrightarrow{\varepsilon} \bigoplus_{i=1}^{4} \mathcal{O}_Z(e_i)^{\oplus n_i+1} \to \mathrm{T}_X|_Z \to 0,$$

whilst the normal bundle sequence is

$$(5)\qquad 0 \to \mathrm{T}_Z \to \mathrm{T}_X|_Z \xrightarrow{\mathrm{d}s} \mathrm{N}_{Z/X} \to 0,$$

where $\mathrm{N}_{Z/X} \cong \mathcal{E}|_Z$, and $\mathrm{d}s$ is induced by the Jacobian. The preceding cohomology computation in Julia gives

(6)

	$\mathrm{T}_X\|_Z$	$\mathrm{N}_{Z/X}$	$\bigwedge^2(\mathrm{T}_X\|_Z)$	$(\mathrm{T}_X\|_Z) \otimes \mathrm{N}_{Z/X}$
h^0	39	39	582	1083
$\mathrm{h}^{\geq 1}$	0	0	0	0

We also have that $\mathrm{h}^0(Z, \mathrm{T}_Z) = \mathrm{h}^1(Z, \mathrm{T}_Z) = u \geq 0$, but we cannot yet determine the value of u. Since Z is Fano, Kodaira–Akizuki–Nakano vanishing gives $\mathrm{h}^{\geq 2}(Z, \mathrm{T}_Z) = 0$.

Vanishing for the tangent bundle For every componentwise nonnegative multidegree m, the Koszul resolution identifies $\mathrm{H}^0(Z, \mathcal{O}_Z(m))$ with $(S/I)_m$ and gives no higher cohomology. Indeed, the twists in this resolution are bounded below componentwise by $(-1,-2,-2,-3)$, so every negative twist lies in the acyclic range of its projective-space factor. The restricted Euler sequence therefore gives $\mathrm{h}^0(Z, \mathrm{T}_X|_Z) = 43 - 4 = 39$ and no higher cohomology.

The normal bundle sequence (5) then gives $u = 39 - \operatorname{rank} \mathrm{H}^0(Z, \mathrm{d}s)$. The following Macaulay2 code continues the earlier snippet, and computes the rank of the Jacobian map from global sections of the middle term in (4) to $\mathrm{H}^0(Z, \mathrm{N}_{Z/X})$. This is the required rank, since those sections surject onto $\mathrm{H}^0(Z, \mathrm{T}_X|_Z)$.

```
R = S/I;
F = R^(degrees S);                  -- middle term of the Euler sequence
N_ZX = R^(degrees source gens I);   -- normal bundle
ds = map(N_ZX, F, substitute(transpose jacobian gens I, R));
A = substitute(matrix entries basis({0,0,0,0}, ds), QQ);
assert((numrows A, numcols A, rank A) == (39,43,39));
```

Thus $u = 0$, proving the cohomology vanishings for the tangent bundle.

Vanishing for the second exterior power of the tangent bundle The second exterior power of the normal bundle sequence (5) gives the exact sequence

$$(7) \qquad 0 \to \bigwedge^2 \mathrm{T}_Z \to \bigwedge^2 (\mathrm{T}_X|_Z) \xrightarrow{\delta} (\mathrm{T}_X|_Z) \otimes \mathrm{N}_{Z/X} \to \operatorname{Sym}^2 \mathrm{N}_{Z/X} \to 0,$$

where $\delta(v \wedge w) = v \otimes \mathrm{d}s(w) - w \otimes \mathrm{d}s(v)$. Taking global sections identifies $\mathrm{H}^0(Z, \bigwedge^2 \mathrm{T}_Z)$ with $\ker \mathrm{H}^0(Z, \delta)$, so it suffices to prove that $\mathrm{H}^0(Z, \delta)$ is injective.

The second exterior power of the Euler sequence (4) and its tensor product with $\mathrm{N}_{Z/X}$ resolve the source and target of δ, respectively, by direct sums of line bundles with nonnegative twists. By the preceding Koszul computation, taking degree zero in their graded presentations therefore computes $\mathrm{H}^0(Z, \delta)$.

The following Macaulay2 code continues the earlier snippets, and constructs this map and checks that its matrix, of size 1083×582 by (6), has rank 582.

```
-- restricted Euler sequence
coordinates = flatten entries vars R;
epsilon = matrix apply(coordinates, x -> apply(degree x, weight -> weight*x));
T_XZ = coker map(F, R^4, epsilon);   -- restricted tangent bundle

-- descend v tensor w |-> v tensor ds(w) - w tensor ds(v)
-- first to wedge^2 F, then to wedge^2(T_X|_Z).
onTensors = F ** ds - flip(N_ZX,F) * (ds ** F);
onWedges = quotient'(onTensors, wedgeProduct(1,1,F));
delta = inducedMap(T_XZ ** N_ZX, exteriorPower(2,T_XZ), onWedges);

-- map on global sections
B = substitute(matrix entries basis({0,0,0,0}, delta), QQ);
assert((numrows B, numcols B, rank B) == (1083,582,582));
```

Thus $\mathrm{H}^0(Z, \delta)$ is injective, proving $\mathrm{H}^0(Z, \bigwedge^2 \mathrm{T}_Z) = 0$. This finishes the proof of Theorem A.

A blowup description We can also give a birational description of Z, by encoding the equations from (1) step-by-step using blowups.

Proposition 1 (Blowup description). *Let $C \subset \mathbb{P}^3_{\mathbf{d}}$ be the twisted cubic defined by the maximal minors of* (10)*, and set $V = \mathrm{Bl}_C \mathbb{P}^3_{\mathbf{d}}$ to be the Fano 3-fold with Mori–Mukai ID* 2.27*. Let $\Gamma \subset \mathbb{P}^2_{\mathbf{a}} \times V$ be the closure of the graph of the rational map on V induced by the projection*

$$(8) \qquad \mathbb{P}^3_{\mathbf{d}} \dashrightarrow \mathbb{P}^2_{\mathbf{a}}, \qquad [d_0 : d_1 : d_2 : d_3] \longmapsto [d_0 : d_3 : d_1 + d_2].$$

Then $\Gamma \cong \mathrm{Bl}_p V$, where $p = [0 : 1 : -1 : 0] \notin C$. With this setup, there exists an isomorphism

$$(9) \qquad Z \cong \mathrm{Bl}_\Gamma(\mathbb{P}^2_{\mathbf{a}} \times V).$$

Proof. The twisted cubic C has a determinantal presentation: its homogeneous ideal is generated by the maximal minors of

$$(10) \qquad \begin{pmatrix} d_0 & d_1 & d_2 \\ d_1 & d_2 & d_3 \end{pmatrix}.$$

By the standard determinantal description of the blowup of a twisted cubic [4, page 686], the last two equations in (1) cut out $V = \mathrm{Bl}_C \mathbb{P}^3_{\mathbf{d}}$. The map (8) is the projection from p. Since $p \notin C$, we identify p with its unique inverse image in V. The graph closure of the induced rational map on V is $\Gamma \cong \mathrm{Bl}_p V$.

Write $(q_0, q_1, q_2) = (d_0, d_3, d_1 + d_2)$. On the chart $a_j \neq 0$, eliminate b_j from the first equation in (1). The second equation then becomes

$$\sum_{i \neq j} (q_i - a_i q_j / a_j) b_i = 0. \tag{11}$$

The two coefficients generate the ideal of Γ on this chart. Since Γ is smooth of codimension 2 in $\mathbb{P}^2_{\mathbf{a}} \times V$, they form a regular sequence, so (11) is the incidence equation for its blowup. These local identifications give $Z \cong \mathrm{Bl}_\Gamma(\mathbb{P}^2_{\mathbf{a}} \times V)$. □

Presumably, this description can also be used to prove Theorem A. For the Hodge numbers this follows from the blowup formula, whilst it is not clear to me how to show the anticanonical bundle is ample from this description. For the computations with the tangent bundle, it is probably feasible to prove that $\mathrm{H}^0(Z, \mathrm{T}_Z) = 0$, and thus $\mathrm{H}^1(Z, \mathrm{T}_Z) = 0$ by either the computation in PartialFlagVarieties.jl or a Grothendieck–Riemann–Roch computation. We have not tried to prove the vanishing of $\mathrm{H}^0(Z, \bigwedge^2 \mathrm{T}_Z)$ using this description.

On LLM usage LLMs were used to speed up the coding process of finding candidates and verifying the vanishing in an explicit example. The blowup description was suggested by the LLM.

References

[1] Pieter Belmans. *A noncommutatively rigid Fano fivefold: computation code.* 2026. URL: `https://github.com/pbelmans/noncommutatively-rigid-fano-fivefold`.

[2] Pieter Belmans, Enrico Fatighenti, and Fabio Tanturri. "Polyvector fields for Fano 3-folds". In: *Math. Z.* 304.1 (2023), Paper No. 12, 30. DOI: `10.1007/s00209-023-03261-2`. MR: 4578397.

[3] Pieter Belmans and Javier Fernández Píriz. *PartialFlagVarieties.jl.* Julia package, version 0.4.0. 2026. URL: `https://github.com/HomogeneousTools/PartialFlagVarieties.jl`.

[4] Lorenzo De Biase, Enrico Fatighenti, and Fabio Tanturri. "Fano 3-folds from homogeneous vector bundles over Grassmannians". In: *Rev. Mat. Complut.* 35.3 (2022), pp. 649–710. DOI: `10.1007/s13163-021-00401-2`. MR: 4482268.

[5] Daniel R. Grayson and Michael E. Stillman. *Macaulay2, a software system for research in algebraic geometry.* URL: `https://macaulay2.com`.

Pieter Belmans, `p.belmans@uu.nl`
Mathematical Institute, Utrecht University, Budapestlaan 6, 3584CD Utrecht, Netherlands